AAEC/E319

AUSTRALIAN ATOMIC ENERGY COMMISSION

RESEARCH ESTABLISHMENT

LUCAS HEIGHTS

SPECT – A FORTRAN PROGRAM FOR THE ANALYSIS OF TECHNETIUM-99m AND OTHER GAMMA SPECTRA IN A RADIOISOTOPE QUALITY CONTROL ENVIRONMENT

by

E.L.R. HETHERINGTON

N. R. WOOD

June 1974

ISBN 0 642 99641 5

AUSTRALIAN ATOMIC ENERGY COMMISSION
RESEARCH ESTABLISHMENT
LUCAS HEIGHTS

SPECT - A FORTRAN PROGRAM FOR THE ANALYSIS OF TECHNETIUM-99m AND OTHER GAMMA SPECTRA IN A RADIOISOTOPE QUALITY CONTROL ENVIRONMENT

by

E.L.R. HETHERINGTON
N.R. WOOD



ABSTRACT

The FORTRAN program SPECT has been written for use in a radioisotope quality control environment to process gamma ray spectra for the estimation of specific impurities in a radiopharmaceutical product. The analysis is performed using a radionuclide library containing details of the principal radioisotopes, expected impurities and known spectrum artifacts. The program provides a plot of the spectrum on which significant peaks are marked.

The activities of identified radioisotopes are calculated from principal emission and decay data contained in the library. The program output lists these activities, together with artifact peaks of known origin and details

(Continued)

of unidentified peaks. Comments to assist the user are also provided.

Since the program requires < 70K of storage, it receives a high priority in the operating system used with the AAEC's IBM360/50 computer. This leads to a short 'turn around' period for the quality control of short lived radiopharmaceuticals.

National Library of Australia card number and ISBN 0 642 99641 5

The following descriptors have been selected from the INIS Thesaurus to describe the subject content of this report for information retrieval purposes. For further details please refer to IAEA–INIS–12 (INIS: Manual for Indexing) and IAEA–INIS–13 (INIS: Thesaurus) published in Vienna by the International Atomic Energy Agency.

ENERGY SPECTRA; ERRORS; FORTRAN; GAMMA SPECTRA; IBM COMPUTERS; IMPURITIES; ISOMERIC NUCLEI; QUALITY CONTROL; QUANTITATIVE CHEMICAL ANALYSIS; RADIOPHARMACEUTICALS; S CODES; TECHNETIUM 99

CONTENTS

1. INTRODUCTION

The determination of radionuclidic purity is an important aspect of radiopharmaceutical quality control, since the presence of impurities affects both radiation dose to the patient and the ultimate image quality. Gamma ray spectrometry, aided by computer analysis of results, is an effective means of making such determinations. The application of this technique to the analysis of technetium-99m, used in the manufacture of radiopharmaceuticals at the AAEC Research Establishment, was described by Boyd, Hetherington & Wood (1971) and Hetherington & Wood (1972). The original program, IMPURE, was used to determine technetium-99m, and the impurities molybdenum-99, iodine-131, iodine-132, ruthenium-103 and lanthanum-140 from counts detected in predetermined analysis channels corresponding to one of the gamma energies of the nuclide. The disadvantages of this approach lay in the failure to obtain an indication of the presence of impurities other than those named, and in the inflexibility of the fixed channel approach in the face of instrument calibration drifts. Experience in the use of this program indicated a need to minimise 'user intervention' and improve the presentation of results, so that staff without computer experience could use the program as an effective quality control aid.

This report describes SPECT, the program developed to overcome the limitations of IMPURE. As the principal radionuclide (usually technetium-99m) and the impurities are known, the spectrum is analysed in terms of a radionuclide library containing all relevant details of these nuclides. Peaks not attributable to members of the library are indicated as being of unknown origin; thus unusual occurrences can be detected. The problem of instrument drifts is overcome by the inclusion of calibration data with each run. The results are presented with interpretations relevant to the quality control situation.

2. OUTLINE OF PROGRAM

The program scans a 512-channel gamma spectrum for significant peaks and, for such peaks, evaluates the peak counts and their positions. An attempt is then made to match the energies of these peaks to the energies of the 'principal' photon emissions given in a library of radioisotope gamma energies and apparent photon energies of artifacts.

The principal gamma ray of a radioisotope is here defined as the largest isolated gamma ray not common to any other member of the library. In a quality control environment this is usually the most abundant gamma ray. The total number of library members, including artifacts, must not exceed twenty

and the total number of peaks for each member of the library must not exceed twenty. The program will not attempt to analyse a spectrum containing more than fifty peaks.

The program lists the radioisotopes present and identifies any other peaks of the same energy (within a set tolerance) as a gamma ray emitted by any one of the radioisotopes considered to be present. An attempt is made to match remaining unidentified peaks with artifact occurrences, such as coincidence peaks that can be present in the particular circumstances. Peaks which cannot be matched in any way are considered as unidentified.

The activities of the identified isotopes are calculated from the principal peak counts and decay scheme details. The activities are corrected for decay to a stated reference time and the percentages of each isotope present at this time are found. These details are then written out together with artifact peaks, for which an activity of NIL is stated. The unidentified peaks are listed. The output also contains notes and comments to assist the user.

The program will loop to analyse any number of spectra provided.

3. DESCRIPTION OF THE PROGRAM

3.1 MAIN

The program is well annotated with comment cards; however, a brief description of the program is given, as certain details may not be obvious from the FORTRAN program itself (Appendix A).

At the commencement of processing a series of arrays is set up. Their contents are listed in Appendix B. The data common to the set of spectra to be analysed is then read in (a typical set for the analysis of technetium-99m is shown in Appendix C and is largely self explanatory).

The calibration energies E1 and E2 should be as widely separated as possible. The starting energy should be set just below the lowest peak energy of interest. Due to the empirical lead attenuation formula, no 'principal' peak of a radioisotope should have an energy of less than 100 keV.

At the statement numbered 1003, the details of the spectrum to be analysed are read in. After the analysis the program returns to this statement, and so there must be as many data cards identifying and qualifying a spectrum as there are spectra to be analysed. These data cards are called "CT,DT" cards. If the spectra are to be read from cards, a "CT,DT" card should precede each spectrum. At the end of the data, execution stops automatically.

In the listing of SPECT in Appendix A, the spectrum to be analysed is read

from a disk data set, set up from all the input spectra by a previous step. The method used is designed to cope with minor paper tape errors. The spectrum is written out on the printer.

Starting at the lowest channel equivalent to the starting energy, the smoothed first derivative of the spectrum at the channel is examined and, if it is greater than 1, it is considered that a peak may be starting. If not, then the next channel is selected and the process repeated (see description of GRAD).

Possible peak beginnings are taken and tested by SIGNF, producing a significance criterion (TT) of 4 or greater for a significant peak. The end of a significant peak is also produced (NEND).

For a significant peak (I), the energy, ENERGY (I), the centre location, PEAK (I,1), the counts in the peak, PEAK (I,2), and the channel nearest the peak centre, CENT(I), are calculated and recorded (see description of EST). A running total of the number of significant peaks is also kept (IP).

Having completed the peak search, the program identifies the radioisotopes present. The criterion for a radioisotope being present is that its major isolated gamma ray is within ± 3 keV of one of the peaks present in the spectrum. The radioisotopes present are listed in the array ISOURC. The peak number considered to be due solely to the particular radioisotope (I) is stored in GAMAY (I) and the peak is labelled as identified. Other peaks thought to be caused by the radioisotopes listed as present are relabelled as identified.

Peaks which are probably due to expected artifacts are relabelled as identified and the artifacts present are added to the list of peak sources present, (ISOURC). The tolerance on the agreement between the spectrum peak and the (apparent) energy of the artifact listed in the library is ± 3 keV for genuine photon peaks, *e.g.* X-rays (IOUT = 0 on the library card), and ± 5 keV for true artifacts, *e.g.* coincidence peaks (IOUT = -1 on the library card).

The activity of each radioisotope identified as present is calculated and written out in microcuries (IOUT = 1 on the library card) or millicuries (IOUT = 2 on the library card). Points to note in this step are:

- EFF (Z) is a function subprogram returning, for an argument Z in MeV, the overall full energy peak efficiency of that energy for the counting geometry used (*e.g.* EFF (Z) = $7.3 \times 10^{-4}\ Z^{-1.18}$ for our system).
- The mass attenuation coefficient (μ/ρ) for lead (RMURO) for energy

Z (MeV) is calculated from an empirically derived formula obtained from a least squares fit of published values of μ/ρ for lead for various energies.

The effective lead thickness is obtained in the following way. The overall detector efficiency is determined for a given counting geometry. A lead shield 4 mm thick is placed around a sample of a low energy gamma emitter, *e.g.* technetium-99m with a known source strength of several tens of millicuries and the sample counted in the same geometry. The effective lead thickness is adjusted until the activity of the low energy gamma emitter calculated to be present by the program agrees with the measured activity of the sample.

- No 'principal' gamma ray as used in calculations can have an energy less than 100 keV.
- Artifact activities are written out as NIL. The energies of unidentified peaks are written out together with the 'significance' of the peak (see description of SIGNF).
- If the calculated activity is less than one microcurie, or the number of counts in the principal peak is less than 100, the possible presence only of the radioisotope is indicated.

A plot of log counts/channel versus channel number is produced by the plotter attached to the computer by the automatic plotting routine CCPLOT (N.R. Wood, AAEC unpublished work). The program now commences to analyse the next spectrum if there is one.

3.2 SUBROUTINES

(i) GRAD (N, SLOPE, I) - GRAD generates a 5-point quadratically smoothed first (I = 1) or second (I = 2) derivative (SLOPE) at a given channel number (N), by the method of convolution integers (Savitsky & Golay 1964).

(ii) SIGNF (N1, I, N2, T) - SIGNF checks the peak (possibly beginning at N1 for 'significance'). For a significant peak I is set to 1 and otherwise is 0. The significant peak ends at N2 and has significance of T. The method is described by Barns (1968). A significant peak is one for which $T \geqslant 4$,

where $$T = \frac{SMAX - SMIN}{\sqrt{SSMAX + SSMIN}}$$

SMAX = maximum positive slope of peak,

SMIN = minimum negative slope of peak,

SSMAX = second derivative at the point of maximum slope,

and SSMIN = second derivative at point of minimum slope.

The first and second derivatives are generated by the subroutine GRAD.

(iii) EST (N1, N2, TOT, CTR, PEAK) - EST estimates the characteristics of the peak between channels N1 and N2. A quadratically smoothed baseline is generated and subtracted from the peak total to give the counts in the peak (TOT). The centre of the peak (CTR) is linearly interpolated from the peak counts and the channel nearest the peak centre is found (PEAK). Visual examination of results of this approach has vindicated the method for the simple spectra usually encountered.

4. ERRORS

4.1 Systematic Errors

The principal systematic errors in the analysis lie in peak height determination and the various corrections applied. Since the results obtained are to be regarded as comparative rather than an absolute determination, a crude estimate of the program accuracy can be made by analysis, in the normal geometry, of a sample of known activity. For millicurie quantities of technetium-99m the accuracy is ± 5%, since the same technique is used for calibration purposes and determination of effective lead thickness. For microcurie levels of molybdenum-99 present in such a sample, errors may be of of the order ± 40%. This does not affect the validity of comparative results.

4.2 Errors Due to the Simple Nature of Program Logic

If the peak used to characterise a library member is common to another member of the library, then errors in estimation will occur. For example, if a sample of technetium-99m contains a significant molybdenum-99 impurity ($\gtrsim$ 1%) there will be a contribution from this source to the 140 keV peak.

If the sample contains an unlisted impurity, with its gamma peaks coinciding with those of library isotopes, its presence may not be detected and significant errors may occur in other activity estimates. Such a situation, however, should not arise in a quality control environment, since the likely impurities would be well known.

5. PRACTICAL USE OF PROGRAM

Appendix C shows an example of the complete card deck as submitted at the AAEC. The spectrum input is from a temporary disk data set which has been loaded with the paper tape data. The program itself is accessed as a load module from a permanent disk data set and, apart from the normal printer output, data is written on a third disk data set. This data is used to produce the plots.

In this form the region of core required on the IBM360/50 at the AAEC for the largest step is < 70K. This allows the job to be run in a class with

a short waiting time.

If the program is run elsewhere, the parts of the program dealing with the reading of spectra from paper tape and the parts dealing with the production of the plots may be removed. The spectra could be read from cards and plots obtained elsewhere.

6. REFERENCES

1. Barns, V. (1968) - IEEE Trans. Nucl. Sci. NS-15 (3) 437-54.
2. Boyd, R.E., Hetherington, E.L.R. & Wood, N.R. (1971) - AAEC/E244.
3. Hetherington, E.L.R. & Wood, N.R. (1972) - AAEC/TM605.
4. Savitsky, A. & Galay, M.J.E. (1964) - Anal. Chem., 36 (8) 1627-39.

APPENDIX A

LISTING OF THE PROGRAM SPECT

```
C*************************************************************************
C
C     THIS PROGRAM ANALYSES ANY NUMBER OF 512 CHANNEL GAMMA SPECTRA IN TERMS
C     OF A (VARIABLE ) GIVEN LIBRARY
C     PEAKS NOT BELONGING TO MEMBERS OF THE LIBRARY IDENTIFIED IN THE SAMPLE
C     ARE INDICATED AS OF UNKNOWN ORIGIN
C     THE SPECTRUM MAY BE ATTENUATED THROUGH LEAD
C     THE COUNT TIME,AND REFERENCE TIME OF RESULT,MAY BE VARIED
C
C*************************************************************************
C
C
      DIMENSION A(512),PEAK(50,2),ENERGY(50),T(50),CENT(50)
      DIMENSION IDENT(50),NAME(20,8),IGAMNO(20),GAMMAS(20,20),ISOURC(20)
     1,DATA(20,2),GAMAY(20),ACT(20),IOUT(20),TITLES(18)
      COMMON A
C
C     THE FOLLOWING 3 CARDS ARE CONCERNED WITH PLOTTING
      CALL GPSEND(1,10)
      DIMENSION X(500)
      DATA TITLES/9*'    ','LOG ','COUN','TS P','ER ','CHAN','NEL ',3*'
     1   '/
C
C     READ IN DATA
      READ(1,28)E1,C1,E2,C2
   28 FORMAT(/4F7.2/)
      READ(1,511) NN,NA,ESTART
  511 FORMAT(22X,I2,27X,I2,20X,F4.0//)
      NTOT=NN+NA
      NNP1=NN+1
      DO 101 I=1,NTOT
      READ(1,102)(NAME(I,J),J=1,8),IGAMNO(I),(DATA(I,K),K=1,2),IOUT(I)
  102 FORMAT(8A4,I2,1X,E7.2,1X,F7.2,I2)
      NM=IGAMNO(I)
      READ(1,103)(GAMMAS(I,J),J=1,NM)
  103 FORMAT(9F8.1)
  101 CONTINUE
      READ(1,73)PBT
   73 FORMAT(/42X,F6.4//)
      TEVPER=(E2-E1)/(C2-C1)
C
C     THE FOLLOWING 3 CARDS CONCERN READING SPECTRA FROM PAPER TAPE
      LOGICAL *1 ERR(8)
      READ(10,469)IF,IL
  469 FORMAT(2A4)
C
C*************************************************************************
C*************************************************************************
C
C     BEGINNING OF PROCESSING LOOP
C
 1003 READ(1,1000,END=1001)CT,DT,(TITLES(I),I=1,9)
 1000 FORMAT(2F6.3,9A4)
C
```

```
C      THE FOLLOWING 20 CARDS CAN BE REMOVED AND INSTRUCTIONS TO READ THE
C      SPECTRUM FROM CARDS SUBSTITUTED
C
C      READ IN SPECTRUM FROM DISK (SET UP ON DISK FROM PAPER TAPE BY A
C      PREVIOUS STEP )
       DO 410 I=1,512
       IF(IL.EQ.0) GOTO 410
       IF(IL.GT.6) GOTO 409
       IF(IF.EQ.0) GOTO 409
       IF(IF.EQ.1) GOTO 411
       READ(10,499)IF,IL,A(I)
  499  FORMAT(2A4,F5.0)
       GOTO 410
  411  READ(10,479)IF,IL,A(I)
  479  FORMAT(2A4,F6.0)
       GOTO 410
  409  READ(10,498)IF,IL,ERR
  498  FORMAT(2A4,8A1)
       WRITE(3,488)I,(ERR(J),J=1,8)
  488  FORMAT(1H ,22HTAPE ERROR IN CHANNEL ,I3,17H  THE TAPE READS ,8A1)
       A(I)=0.0
  410  CONTINUE
C
C      WRITE SPECTRUM
       WRITE(3,79)A
   79  FORMAT(1H ,10F10.0)
C*******************************************************************************
C
C      START PEAK SEARCH
       IP=0
       NEND=0
       NSTART=(ESTART-E1)/TEVPER+C1
     7 NEW=MAX0(NSTART,NEND)
       N=NEW-1
     3 N=N+1
       IF(N-505)1,1,2
     1 CALL GRAD(N,SLOPE,1)
       IF(SLOPE-1.)3,3,4
C
C      TEST FOR POSSIBLE PEAK
     4 CALL SIGNF(N,I,NEND,TT)
       IF(NEND-513)621,2,2
  621  CONTINUE
       IF(I)3,3,5
C
C      IF SIGNIFICANT EVALUATE DETAILS
     5 IP=IP+1
       IF(IP-50) 200,201,201
  201  WRITE(3,203)
  203  FORMAT('1THE SPECTRUM IS TOO COMPLEX FOR THE PROGRAM '/'0THE NEXT
      1 SPECTRUM IS NOW EXAMINED IF THERE IS ONE')
       GOTO 1003
  200  CONTINUE
       T(IP)=TT
       CALL EST(N,NEND,PEAK(IP,1),PEAK(IP,2),CENT(IP))
       EE1=(PEAK(IP,1)-C1)*TEVPER+E1
       EE2=-(C2-PEAK(IP,1))*TEVPER+E2
       ENERGY(IP)=(EE1+EE2)/2.
       GO TO 7
     2 CONTINUE
```

```
C*************************************************************************
C
C     LIST WHICH ISOTOPES ARE PRESENT ( MAJOR GAMMA EXISTS IN SPECTRUM )
C     REMEMBERING THE NUMBER OF THE PEAK ASSOCIATED WITH THE PRINCIPLE
C     GAMMA RAY OF EACH ISOTOPE PRESENT.
      NGAM=0
      DO 42 I=1,50
   42 IDENT(I)=0
      DO 41 I=1,IP
      DO 43 IS=1,NN
      IF(ABS(ENERGY(I)-GAMMAS(IS,1))-3.)45,43,43
   45 IDENT(I)=IS
      NGAM=NGAM+1
      ISOURC(NGAM)=IS
      GAMAY(IS)=I
      GOTO 41
   43 CONTINUE
   41 CONTINUE
C
C     ELIMINATE OTHER GAMMAS OF IDENTIFIED ISOTOPES
      DO 66 IK=1,IP
      IF(IDENT(IK).GT.0) GOTO 66
      DO 60 I=1,NGAM
      ISO=ISOURC(I)
      IN=IGAMNO(ISO)
      DO 62 IU=1,IN
      IF(ABS(ENERGY(IK)-GAMMAS(ISO,IU))-3.)65,62,62
   65 IDENT(IK)=ISO
      GOTO 66
   62 CONTINUE
   60 CONTINUE
   66 CONTINUE
C
C     ELIMINATE PEAKS DUE TO EXPECTED ARTIFACTS EXTENDING THE LIST OF PEAK
C     SOURCES PRESENT.
      DO 120 I=1,IP
      IF(IDENT(I).GT.0) GOTO 120
      DO 121 J=NNP1,NTOT
      GAP=3.0
      IF(IOUT(J).LT.0) GAP=5.
      IF(ABS(ENERGY(I)-GAMMAS(J,1))-GAP)122,121,121
  122 IDENT(I)=J
      NGAM=NGAM+1
      ISOURC(NGAM)=J
      GOTO 120
  121 CONTINUE
  120 CONTINUE
C*************************************************************************
C
C     ACTIVITY CALCULATION
C
C     DUE TO POOR RESOLUTION ACTIVITY IS ONLY CALCULATED ON THE LARGEST
C     ISOLATED PEAK OF THE ISOTOPE (ENERGY >100KEV)
C
      DO 91 I=1,NGAM
      ISO=ISOURC(I)
      IF(IOUT(ISO))91,91,82
   82 NO=GAMAY(ISO)
      Z=GAMMAS(ISO,1)/1000.
      IF(Z.LT.0.1) GOTO 94
```

```
      RMURO=10.**(1.0999*(ALOG10(Z))**2-0.8233*ALOG10(Z)-1.1809)
      CM=PEAK(NO,2)/EFF(Z)/EXP(-PBT*11.3437*RMURO)/DATA(ISO,1)
      CM=CM*0.69315/(1.-EXP(-0.69315*CT/DATA(ISO,2)))/DATA(ISO,2)*EXP(0.
     169315*DT/DATA(ISO,2))
      ACT(ISO)=CM/(3.7E4*3600.)
      GOTO 91
   94 WRITE(3,95)(NAME(ISO,J),J=1,8)
   95 FORMAT('0',8A4,' HAS A "PRINCIPLE" GAMMA RAY <100KEV ,'/' IT IS PR
     1ESENT BUT IGNORED IN FUTHER CALCULATIONS')
   91 CONTINUE
      TOT=0.0
      DO 85 I=1,NGAM
      ISO=ISOURC(I)
      IF(IOUT(ISO))85,85,92
   92 TOT=TOT+ACT(ISO)
   85 CONTINUE
C
C     OUTPUT
C
      WRITE(3,72)(TITLES(I),I=1,9),CT,DT
   72 FORMAT('1'////' SAMPLE IDENTIFICATION  ',9A4,'  COUNT TIME = ',F6.
     13,' HRS        DECAY TIME = ',F6.3,' HRS'////)
      WRITE(3,71)
   71 FORMAT('0IDENTIFIABLE RADIOISOTOPES OR                ACTIVITY
     1          PERCENTAGE OF IDENTIFIABLE                          '/' ',
     26X,'PEAK SOURCES',22X,' (MICROCURIES) ',15X,'GAMMA ACTIVITY'/' ',6
     35X,'(COMPARATIVE PURPOSES ONLY)'/)
C
C     IDENTIFIED PEAK SOURCES GIVEN
      DO 83 I=1,NGAM
      ISO=ISOURC(I)
      IF(IOUT(ISO))87,87,89
   87 WRITE(3,90)(NAME(ISO,J),J=1,8)
   90 FORMAT('0',8A4,14X,'NIL')
      GOTO 83
   89 PC=ACT(ISO)/TOT*100.0
      IX=GAMAY(ISO)
      SD=100.0/SQRT(PEAK(IX,2))
      IF((SD.GT.10.).OR.(ACT(ISO).LT.1.)) GOTO 88
      IF(IOUT(ISO).EQ.2) ACT(ISO)=ACT(ISO)/1000.
      IACT=ACT(ISO)+0.5
      WRITE(3,86)(NAME(ISO,J),J=1,8),IACT,PC
   86 FORMAT('0',8A4,11X,I6,26X,F8.4)
      GOTO 83
   88 WRITE(3,93)(NAME(ISO,J),J=1,8)
   93 FORMAT('0',8A4,3X,'POSSIBLE TRACE DETECTED')
   83 CONTINUE
C
C     UNIDENTIFIED PEAK ENERGIES GIVEN
      WRITE(3,300)
  300 FORMAT(' '////' UNIDENTIFIED PEAK ENERGIES',10X,'PEAK SIGNIFICANCE
     1'/' ',5X,'+OR- 3 KEV'/)
      IQ=0
      DO 301 I=1,IP
      IF(IDENT(I))302,302,301
  302 IE=ENERGY(I)+0.5
      WRITE(3,304)IE,T(I)
  304 FORMAT(' ',10X,I4,26X,F5.1)
      IQ=IQ+1
  301 CONTINUE
```

```
      IF(IQ)305,305,306
  305 WRITE(3,307)
  307 FORMAT(' ',8X,'NIL')
  306 CONTINUE
C
C     WARNING OF POSSIBLE INVALID ANALYSIS
      IF(IQ-4)206,207,207
  207 WRITE(3,208)
  208 FORMAT(//' ************************************************************
     1'/' *  THE LIST OF UNIDENTIFIED PEAKS IS ABNORMALLY LONG  *'/' *
     2THE ANALYSIS MUST BE INCOMPLETE OR WRONG            *'/' **********
     3*******************************************')
  206 CONTINUE
C
C     DUE TO THE SIMPLE NATURE OF THE PROGRAM A VISUAL INSPECTION OF THE GAMMA
C     SPECTRUM IS ALWAYS ADVISABLE
      WRITE(3,310)
  310 FORMAT(' '//////' CAUTION:   THE ABOVE RESULTS SHOULD BE  ACCOMPAN
     1IED BY A SENSIBLE SPECTRUM PLOT')
      WRITE(3,876)
  876 FORMAT('1')
C
C
C     THE FOLLOWING 13 CARDS GENERATE THE PLOT AND MAY BE REMOVED
      DO 500 I=1,500
      X(I)=FLOAT(I)
      IF(A(I))501,501,502
  502 A(I)=ALOG10(A(I))
      GOTO 500
  501 A(I)=0.0
  500 CONTINUE
      CALL CCPLOT(X,A,500,9.5,6.75,TITLES,1,0,1,0,IR)
      DO 503 I=1,IP
      X(I)=PEAK(I,1)
  503 A(I)=ALOG10(CENT(I)*1.0)
C
      CALL CCPLOT(X,A,IP,9.5,6.75,TITLES,0,1,0,3,IR)
      GO TO 1003
 1001 CONTINUE
C     THE FOLLOWING CARD IS CONCERNED WITH PLOTTING
      CALL GPSEND(2)
C
      STOP
      END



      FUNCTION EFF(X)
      EFF=7.3E-4/X**1.18
      RETURN
      END
```

```
      SUBROUTINE EST(N1,N2,CTR,TOT,PEAK)
C     ESTIMATES CHARACTERISTICS OF PEAK BETWEEN N1 AND N2
      DIMENSION A(512)
      COMMON A
      Y1=-3.*A(N1-2)+12.*A(N1-1)+17.*A(N1)+12.*A(N1+1)-3.*A(N1+2)
      Y1=Y1/35.
      Y2=-3.*A(N2-2)+12.*A(N2-1)+17.*A(N2)+12.*A(N2+1)-3.*A(N2+2)
      Y2=Y2/35.
      N=N2-N1
      BGTOT=(Y1+Y2)/2.*FLOAT(N)
      TOT=0.0
      BIG=0.0
      DO 1 I=N1,N2
      TOT=TOT+A(I)
      BIG=AMAX1(BIG,A(I))
      IF(BIG-A(I))2,2,1
    2 J=I
    1 CONTINUE
      PEAK=A(J)
      K=J-2
      L=J+2
      TOT=TOT-BGTOT
      T1=0.0
      T2=0.0
      DO 6 I=K,L
      T1=T1+A(I)
    6 T2=T2+A(I)*FLOAT(I)
      CTR=T2/T1
      RETURN
      END
```

```
      SUBROUTINE SIGNF(N1,I,N2,T)
C     CHECKS PEAK ( POSSIBLY BEGINNING AT N1 ) FOR SIGNIFICANCE
      DIMENSION A(512)
      COMMON A
      N=N1
      CALL GRAD(N,G1,1)
      NMAX=N
      SMAX=G1
    1 N=N+1
      IF(N-512)6,5,5
    6 CALL GRAD(N,G2,1)
      IF(G2)2,2,3
    3 IF(G1-G2)4,1,1
    4 SMAX=G2
      NMAX=N
      GO TO 1
    2 G1=G2
      NMIN=N
      SMIN=G2
    7 N=N+1
      IF(N-512)13,5,5
   13 CALL GRAD(N,G2,1)
      IF(G2+((-3.*A(N-2)+12.*A(N-1)+17.*A(N)+12.*A(N+1)-3.*A(N+2))/(35.*
     120.)))8,9,9
    8 IF(G2-G1)10,7,7
   10 SMIN=G2
      NMIN=N
      G1=G2
      GOTO 7
    9 CONTINUE
      N2=N
      CALL GRAD(NMAX,SSMAX,2)
      CALL GRAD(NMIN,SSMIN,2)
      D=SMAX-SMIN
      S=SQRT(SSMAX+SSMIN)
      IF(S)11,11,77
   77 CONTINUE
      T=D/S
      IF(T-4.0)11,12,12
   12 I=1
      RETURN
   11 I=0
      RETURN
    5 WRITE(3,14)N1
   14 FORMAT(' THE PEAK BEGINNING IN CHANNEL ',I3,' ENDS OUTSIDE THE SPE
     1CTRUM')
      N2=513
      RETURN
      END

      SUBROUTINE GRAD(N,SLOPE,I)
C     GENERATES 5 POINT QUADRATICALLY SMOOTHED FIRST OR SECOND DERATIVE AT N
      DIMENSION A(512)
      COMMON A
      SLOPE=0.0
      J=-3
    3 J=J+1
      M=N+J
      SLOPE=A(M)*((FLOAT(J)/10.0)**I)+SLOPE
      IF(J-2)3,1,1
    1 RETURN
      END
```

APPENDIX B

DETAILS OF ARRAYS, CONSTANTS AND SUBROUTINES USED BY SPECT

B1. CONTENTS OF ARRAYS

A	-	gamma spectrum to be analysed (512 channels)
PEAK (I,1)	-	location of peak I
PEAK (I,2)	-	counts in peak I
ENERGY (I)	-	gamma energy of peak I in keV
T (I)	-	statistical significance of peak I (> 4 implies peak is significant)
*CENT (I)	-	channel closest to centre of peak I
IDENT (I)	-	library number of gamma emitter considered to be associated with peak I
NAME (I,J)	-	J = 1, 8 - 32 character name of gamma emitter I
IGAMMO (I)	-	number of gamma rays emitted by radioisotope I
GAMMAS (I,J)	-	the energies of the J gamma rays emitted by isotope I
ISOURC (I)	-	list of the numbers of the gamma sources present in ascending order
GAMAY (I)	-	number of the peak which is the principal peak of gamma source I
ACT (I)	-	gamma activity of gamma source I
IOUT (I)	-	extra information about library member I
= 2	-	activity output is in millicuries
= 1	-	activity output is in microcuries
= 0	-	gamma peak with nil activity (*e.g.* X-ray)
= -1	-	artifact peak with nil activity (*e.g.* coincidence peak)
DATA (I,1)	-	abundance of principal gamma ray of gamma source I
DATA (I,2)	-	half life (hours) of gamma source I
*TITLES (I)	-	I = 1, 9 - 36 character list identifying spectrum being analysed and is also X-axis plot label I = 10, 18 - 36 character list which is y axis plot label
*X	-	contains the channel numbers 1 to 500

(* concerned with plotting)

B2. IMPORTANT CONSTANTS

E1	-	calibration energy near beginning of spectrum in keV
C1	-	channel in which E1 occurs
E2	-	calibration energy near end of spectrum in keV
C2	-	channel in which E2 occurs
NN	-	number of radioisotopes in the library
NA	-	number of artifact sources in the library
PBT	-	thickness of lead around sample in cm
CT	-	count time of spectrum in hours
DT	-	reference time of result relative to the time the spectrum was made, in hours + if before spectrum was made - if after spectrum was made
NGAM	-	Number of library members present in spectrum
ESTART	-	energy at which spectrum search is to begin in keV
RMURO	-	μ/ρ for lead for a gamma ray of energy Z MeV

B3. FUNCTION SUBPROGRAM

EFF (Z)	-	Overall full energy peak detector efficiency formula (Must be determined for each detector system)

B4. SUBROUTINES

GRAD	-	First and second derivative of a tabulated function
SIGNF	-	Significance of a spectrum peak
EST	-	Characteristics of a spectrum peak

APPENDIX C

A TYPICAL JOB SUBMITTED AT THE AAEC RESEARCH ESTABLISHMENT

```
//NRWSPECT JOB (I1C21938,D3),N.R.WOOD,
//             CLASS=A,
// TIME=2
//PLOT   EXEC  BUFFPROG,PRG=AEPLOT
//PTREC  EXEC  PGM=PTREC
//SYSPRINT  DD  SYSOUT=A
//SYSUT1  DD  UNIT=PTAPE
//STEPLIB  DD  DSN=GWC.LINKLIB,DISP=SHR
//SYSUDUMP  DD  SYSOUT=A
//SYSUT2  DD  UNIT=SYSDA,DISP=(,PASS),DCB=(RECFM=FB,LRECL=78,
//  BLKSIZE=780),SPACE=(TRK,(2,2))
//SYSIN  DD  *
 CODE ASCII IGNORE 80,00,8A,0A,FF, TERMINATOR 8D, 8D, A0, 20.
PRINT OFF
 FORMATS
(F6)
(F5)
/*
//  EXEC  FORTHG,DSN='ISOTOPE.NRW',PRG=SPECT4,REGION.GO=70K
//GO.AEPLOT DD SYSOUT=C
//GO.FT10F001  DD  DSN=*.PTREC.SYSUT2,DISP=(OLD,PASS)
//GO.SYSIN  DD  *
THE CALIBRATION CARD FOLLOWS AS E1,C1,E2,C2 WITH A FORMAT OF 4F7.2
  59.6   24.801173.2  478.2

NO OF RADIOISOTOPES = 08  NO OF ARTIFACT SOURCES = 05  STARTING ENERGY = 0065

THE LIBRARY FOLLOWS
TECHNETIUM 99M ( MILLICURIES )   2 9.00E-1    6.00 2
   140.00  322.0
IODINE 131 (I132 MAY INTERFER)   3 8.00E-2  193.20 1
   637.0   364.5   284.3
IODINE 132                      15 1.04E+0    2.30 1
   668.0   523.0   630.0   773.0   955.0   813.0  1035.0   728.0   810.0
  1135.5   505.7   621.0   877.0   650.6   911.0
MOLYBDENUM 99                    6 1.20E-1   66.70 1
   740.0   823.0   961.0   778.0   366.4   181.1
LANTHANUM 140                    2 4.65E-1   40.22 1
   487.0   816.0
RUTHENIUM 103                    1 9.00E-1  950.40 1
   497.1
RHENIUM 188                      4 6.00E-3   16.70 1
   478.0   633.0   829.0   932.0
RHENIUM 186                      1 3.00E-4   90.00 1
   767.0
LEAD K ALPHA X-RAY               1                0
    74.0
LEAD K BETA X-RAY                1                0
    86.0
PB  K ALPHA,TC99M COINCIDENCE    1               -1
   214.0
PB  K BETA,TC99M COINCIDENCE     1               -1

   226.0
TC99M,TC99M COINCIDENCE          1               -1
   280.0

THE EFFECTIVE LEAD SHIELDING THICKNESS IS 0.4074 CMS

THE "CT,DT" CARD(S) FOLLOW  WITH A FORMAT OF 2F6.3,9A4
 0.278 1.000 TC99M FROM (N,GAMMA) PROD MO99
 0.278 0.300 TC99M FROM FISSION PROD MO99
/*
//
//
```

SAMPLE IDENTIFICATION TC99M FROM (N,GAMMA) PROD MO99 COUNT TIME = 0.278 HRS DECAY TIME = 1.000 HRS

IDENTIFIABLE RADIOISOTOPES OR PEAK SOURCES	ACTIVITY (MICROCURIES)	PERCENTAGE OF IDENTIFIABLE GAMMA ACTIVITY (COMPARATIVE PURPOSES ONLY)
TECHNETIUM 99M (MILLICURIES)	277	99.9999
IODINE 132	POSSIBLE TRACE DETECTED	
LEAD K ALPHA X-RAY	NIL	
LEAD K BETA X-RAY	NIL	

UNIDENTIFIED PEAK ENERGIES +OR- 3 KEV	PEAK SIGNIFICANCE
272	7.6

CAUTION THE ABOVE RESULTS SHOULD BE ACCOMPANIED BY A SENSIBLE SPECTRUM PLOT


LOG COUNTS PER CHANNEL
TC99M FROM (N,GAMMA) PROD M099

SAMPLE IDENTIFICATION TC99M FROM FISSION PROD MO99 COUNT TIME = 0.278 HRS DECAY TIME = 0.300 HRS

IDENTIFIABLE RADICISOTOPES OR PEAK SOURCES	ACTIVITY (MICROCURIES)	PERCENTAGE OF IDENTIFIABLE GAMMA ACTIVITY (COMPARATIVE PURPOSES ONLY)
TECHNETIUM 99M (MILLICURIES)	154	99.9975
IODINE 132	2	0.0016
MOLYBDENUM 99	1	0.0009
LEAD K ALPHA X-RAY	NIL	
LEAD K BETA X-RAY	NIL	

UNIDENTIFIED PEAK ENERGIES +OR- 3 KEV	PEAK SIGNIFICANCE
1175	4.9

CAUTION THE ABOVE RESULTS SHOULD BE ACCOMPANIED BY A SENSIBLE SPECTRUM PLOT


TC99M FROM FISSION PROD MO99
LOG COUNTS PER CHANNEL